

Four Problems with Policy-Based Constraints and How to Fix Them

Dillan Mills
Synopsys, Inc.
Maricopa, AZ
dillan@synopsys.com

Chip Haldane
The Chip Abides, LLC
Gilbert, AZ
chip@thechipabides.com

Abstract

This paper presents solutions to problems encountered in the implementation of policy classes for SystemVerilog constraint layering. Policy classes provide portable and reusable constraints that can be mixed and matched into the object being randomized. There have been many papers and presentations on policy classes since the original presentation by John Dickol at DVCon 2015. The paper addresses three problems shared by all public policy class implementations and presents a solution to a fourth problem. The proposed solutions introduce policy class inheritance, tightly pair policy definitions with the class they constrain, reduce the expense of defining common policies using macros, and demonstrate how to treat policies as disposable and lightweight objects. The paper concludes that the proposed solution improves the usability and efficiency of policy classes for SystemVerilog constraint layering.

I. CONSTRAINTS AND POLICY CLASS REVIEW

When working with randomized objects in SystemVerilog, it is typically desired to control the randomization of the item being randomized. The common method of doing so is by using constrained random value generation. Constraints for the randomization of class properties can be provided within the class definition, which will require randomized properties to always pass the specified constraint check. For example, the following `addr_txn` class contains a constraint on the `addr` property to ensure it is always word aligned.

```
1 class addr_txn;
2     rand bit [31:0] addr;
3     rand int      size;
4
5     constraint c_size {size inside {1, 2, 4}};
6 endclass
```

Constraints can also be provided outside the class definition at the point of randomization by using the `with` construct. This allows variation to the constraints based on the context of the randomization. For example, the same `addr_txn` class can have an instance of it randomized with the following constraint on its `data` property.

```
1 addr_txn txn = new();
2 txn.randomize() with {addr == 32'hdeadbeef};
```

Policy classes are a technique for SystemVerilog constraint layering with comparable performance to traditional constraint methods, while additionally providing portable and reusable constraints that can be mixed and matched into the object being randomized, originally presented by John Dickol [1] [2]. This is possible through SystemVerilog “Global Constraints”, which randomizes all lower-level objects whenever a top-level object is randomized, and solves all rand variables and constraints across the entire object hierarchy simultaneously. A summary of his approach can be found by examining the following code.

```
1 class policy_base#(type ITEM=uvm_object);
2     ITEM item;
3
4     virtual function void set_item(ITEM item);
```

```

5         this.item = item;
6     endfunction
7 endclass
8
9
10 class policy_list#(type ITEM=uvm_object) extends policy_base#(ITEM);
11     rand policy_base#(ITEM) policy[$];
12
13     function void add(policy_base#(ITEM) pcy);
14         policy.push_back(pcy);
15     endfunction
16
17     function void set_item(ITEM item);
18         foreach(policy[i]) policy[i].set_item(item);
19     endfunction
20 endclass

```

These first two classes provide the base class types for the actual policies. This allows different, related policies to be grouped together in **policy_list** and applied to the randomization call concurrently. Both of these classes are parameterized to a specific object type, so separate instances will be required for each class type that needs to be constrained.

```

1 class addr_policy_base extends policy_base#(addr_txn);
2     addr_range ranges[$];
3
4     function void add(addr_t min, addr_t max);
5         addr_range rng = new(min, max);
6         ranges.push_back(rng);
7     endfunction
8 endclass
9
10
11 class addr_permit_policy extends addr_policy_base;
12     rand int selection;
13
14     constraint c_addr_permit {
15         selection inside {[0:ranges.size()-1]};
16
17         foreach(ranges[i]) {
18             if(selection == i) {
19                 item.addr inside {[ranges[i].min:ranges[i].max - item.size]};
20             }
21         }
22     }
23 endclass
24
25
26 class addr_prohibit_policy extends addr_policy_base;
27     constraint c_addr_prohibit {
28         foreach(ranges[i]) {
29             !(item.addr inside {[ranges[i].min:ranges[i].max - item.size + 1]});
30         }
31     }
32 endclass

```

The next three classes contain implementations of the `policy_base` class for the `addr_txn` class. A list of valid address ranges can be stored in the `ranges` array. The `addr_permit_policy` will choose one of the ranges at random and constrain the address to be within the range, while the `addr_prohibit_policy` will exclude the address from randomizing inside any of the ranges in its list.

```
1 class addr_txn;
2     rand bit [31:0] addr;
3     rand int      size;
4     rand policy_base#(addr_txn) policy[$];
5
6     constraint c_size {size inside {1, 2, 4};}
7
8     function void pre_randomize;
9         foreach(policy[i]) policy[i].set_item(this);
10    endfunction
11 endclass
12
13
14 class addr_constrained_txn extends addr_txn;
15     function new;
16         addr_permit_policy permit = new;
17         addr_prohibit_policy prohibit = new;
18         policy_list#(addr_txn) pcy = new;
19
20         permit.add('h00000000, 'h0000FFFF);
21         permit.add('h10000000, 'h1FFFFFFF);
22         pcy.add(permit);
23
24         prohibit.add('h13000000, 'h130FFFFF);
25         pcy.add(prohibit);
26
27         this.policy = {pcy};
28     endfunction
29 endclass
```

The final two classes show how these policies are used. The `addr_txn` class creates a `policy` queue, and during the `pre_randomize` function sets the handle to the `item` object within each policy to itself. The `addr_constrained_txn` subclass adds two policies to a local `pcy` list, one that permits the address to be within one of two ranges, and one that prohibits the address from being within a third range. The `addr_constrained_txn` class can then pass the local `pcy` list to the parent `policy` queue.

At this point, an instance of `addr_constrained_txn` can be created and randomized like normal, and the address will be constrained based on the policies provided.

```
1 addr_constrained_txn txn = new();
2 txn.randomize();
```

Further work on policy-based constraints has been presented since the original DVCon presentation in 2015. Kevin Vasconcellos and Jeff McNeal applied the concept to test configuration and added many nice utilities to the base policy class [3]. Chuck McClish extended the concept to manage real number values for User Defined Nettypes (UDN) and Unified Power Format (UPF) pins in an analog model [4]. Additionally, he defined a policy builder class that was used to generically build multiple types of policies while reducing repeated code that was shared between each policy class in the original implementation.

Although there has been extensive research on policies, this paper aims to address and provide solutions for three issues that have not been adequately resolved in previous implementations. Furthermore, a fourth problem that arose during testing

of the upgraded policy package implementation will also be discussed and resolved.

II. PROBLEM #1: PARAMETERIZED POLICIES

The first problem with the above policy implementation is that because policies are specialized using parameterization to the class they constrain, different specializations of parameterized classes cannot be grouped and indexed. This is most apparent when creating a class hierarchy. If you extend `addr_txn` and add additional fields to be constrained, then you need a new policy type. The new policy type requires an extra policy list to modify any new attributes in the child class and requires the user to know which layer of the class hierarchy defines each attribute they want to constrain. Imagine a complex class hierarchy with many layers of inheritance and extension, and new constrainable attributes on each layer (one common example is a multi-layered sequence API library, such as the example presented by Jeff Vance [5]). Using policies becomes cumbersome in this case because you need to manage a separate policy list for each level of the hierarchy and know what each list can and cannot contain. You could end up with a final class that looks like this:

```
1  class addr_l2_txn extends addr_l1_txn;
2      rand int f2;
3      rand policy_base#(addr_l2_txn) policy_l2[$];
4
5      function void pre_randomize;
6          super.pre_randomize();
7          foreach(policy_l2[i]) policy_l2[i].set_item(this);
8      endfunction
9  endclass
10
11  // ... repeat for addr_l3_txn and addr_l4_txn
12
13  class addr_constrained_txn extends addr_l4_txn;
14      function new;
15          addr_permit_policy      permit = new;
16          addr_prohibit_policy    prohibit = new;
17          addr_l2_policy          l2 = new;
18          addr_l3_policy          l3 = new;
19          addr_l4_policy          l4 = new;
20          policy_list#(addr_l1_txn) pcy_l1 = new;
21          policy_list#(addr_l2_txn) pcy_l2 = new;
22          policy_list#(addr_l3_txn) pcy_l3 = new;
23          policy_list#(addr_l4_txn) pcy_l4 = new;
24
25          permit.add('h00000000, 'h0000FFFF);
26          permit.add('h10000000, 'h1FFFFFFF);
27          pcy_l1.add(permit);
28          prohibit.add('h13000000, 'h130FFFFF);
29          pcy_l1.add(prohibit);
30
31          l2.set('h12345678); pcy_l2.add(l2);
32          l3.set('h55555555); pcy_l3.add(l3);
33          l4.set('h87654321); pcy_l4.add(l4);
34
35          this.policy_l1 = {pcy_l1};
36          this.policy_l2 = {pcy_l2};
37          this.policy_l3 = {pcy_l3};
38          this.policy_l4 = {pcy_l4};
39      endfunction
40  endclass
```

This class ends up with a lot of boilerplate code and is not very intuitive to use. The user needs to keep track of what the difference is between each policy list and apply constraints properly. Additionally, more inheritance causes the class to grow further. This would strongly discourage using inheritance, which is not desirable.

The solution to this problem is to introduce class inheritance into the policy classes that mirror the target class hierarchy. So for the four-layered `addr_txn` example, four matching policy classes would be created to match.

```
1  class addr_l1_policy extends policy_base#(addr_l1_txn);
2      addr_range ranges[$];
3
4      function add(addr_t min, addr_t max);
5          addr_range rng = new(min, max);
6          ranges.push_back(rng);
7      endfunction
8  endclass
9
10 class addr_l2_policy extends addr_l1_policy;
11     // ... additional fields that are used to constrain the object
12 endclass
13
14 class addr_l3_policy extends addr_l2_policy;
15     // ... additional fields that are used to constrain the object
16 endclass
17
18 class addr_l4_policy extends addr_l3_policy;
19     // ... additional fields that are used to constrain the object
20 endclass
```

, supplemented by convenient macros to keep the solution simple.

III. PROBLEM #2: DEFINITION LOCATION

The second problem is “where do I define my policy classes?” This is not a complicated problem to solve, and most users likely place their policy classes in a file close to the class they are constraining. However, this paper will present a solution that explicitly pairs policy definitions with the class they constrain, so users never need to hunt for the definitions or guess where they are located, and so the definitions have an appropriate and consistent home in the environment. This is done by embedding a **POLICIES** class inside the class being constrained. This class acts as a container for all relevant policies and allows policies to be created by simply referencing the embedded class through the `::|` operator.

IV. PROBLEM #3: BOILERPLATE OVERLOAD

The third problem is that policies are relatively expensive to define since you need at a minimum, a class definition, a constructor, and a constraint. This will be relatively unavoidable for complex policies, such as those defining a relationship between multiple specific class attributes. For generic policies, such as equality constraints, range constraints, or set membership (keyword **inside**) constraints, a strategy using macros is presented that drastically reduces the expense and risk of defining common policies. The macros are responsible for creating the specialized policy class for the required constraint within the target class, as well as a static constructor function that is used to create new policy instances of the class. Additional macros are utilized for setting up the embedded **POLICIES** class within the target class. These macros can be used hand in hand with the non-macro policy classes needed for the complex constraints, if necessary.

V. PROBLEM #4: UNEXPECTED POLICY REUSE BEHAVIOR

A fourth problem we noticed in our usage of policy classes was occasional unexpected behavior when attempting to reuse a policy. It appeared as though the policies “remembered” being randomized and wouldn’t reapply their constraints during subsequent **randomize** calls. An exact diagnosis of the issue was not discovered because we transitioned our policy approach to a “use once and discard” approach, where we would treat a policy as disposable and model them as extremely lightweight classes. We implemented practices such as static functions to create new instances of the policies and policy methods to return fresh copies of themselves.

VI. CONCLUSION

Functional example code will be provided to demonstrate the solutions to these problems and provide other minor improvements to the original policy class implementation.

APPENDIX A POLICY PACKAGE

A. *policy_pkg.sv*

```
1  `ifndef __POLICY_PKG__
2  `define __POLICY_PKG__
3
4  `include "uvm_macros.svh"
5
6  package policy_pkg;
7
8      import uvm_pkg::*;
9
10     `include "policy.svh"
11
12     typedef policy POLICY_QUEUE[$];
13
14     `include "policy_container.svh"
15
16     `include "policy_imp.svh"
17     `include "policy_list.svh"
18
19 endpackage: policy_pkg
20
21 `endif
```

B. *policy.svh*

```
1  `ifndef __POLICY__
2  `define __POLICY__
3
4  interface class policy;
5
6      pure virtual function string name();
7
8      pure virtual function string type_name();
9
10     pure virtual function string description();
11
12     pure virtual function bit item_is_compatible(uvm_object item);
13
14     pure virtual function void set_item(uvm_object item);
15
16     pure virtual function policy copy();
17
18 endclass: policy
19
20 `endif
```

C. *policy_container.svh*

```
1  `ifndef __POLICY_CONTAINER__
2  `define __POLICY_CONTAINER__
3
4  interface class policy_container;
5
6      // Queries
7      pure virtual function bit has_policies();
8
9      // Assignments
10     pure virtual function void set_policies(policy_queue policies);
11
12     pure virtual function void add_policies(policy_queue policies);
13
14     pure virtual function void clear_policies();
15
16     // Access
17     pure virtual function policy_queue get_policies();
18
19     // Copy
20     pure virtual function policy_queue copy_policies();
21
22
23 endclass: policy_container
24
25 `endif
```

D. *policy_imp.svh*

```
1  `ifndef __POLICY_IMP__
2  `define __POLICY_IMP__
3
4  virtual class policy_imp #(type ITEM=uvm_object) implements policy;
5
6      protected rand ITEM m_item;
7
8      pure virtual function string name();
9      pure virtual function string description();
10     pure virtual function policy copy();
11
12     virtual function string type_name();
13         return( ITEM::type_name );
14     endfunction: type_name
15
16     virtual function bit item_is_compatible(uvm_object item);
17         ITEM local_item;
18
19         if (item == null) return( 0 );
20         else return( $cast(local_item, item) );
21     endfunction: item_is_compatible
22
23     virtual function void set_item(uvm_object item);
```

```

24         if (item == null) begin
25             `uvm_error("policy::set_item()", "NULL item passed")
26
27         end else if ((this.item_is_compatible(item)) && $cast(this.m_item, item))
28         begin
29             `uvm_info(
30                 "policy::set_item()",
31                 $sformatf(
32                     "policy <%s> applied to item <%s>: %s",
33                     this.name(), item.get_name(), this.description
34                 ),
35                 UVM_FULL
36             )
37             this.m_item.rand_mode( 1 );
38
39         end else begin
40             `uvm_warning(
41                 "policy::set_item()",
42                 $sformatf(
43                     "Item <%s> type <%s> is not compatible with policy <%s> type
44                     <%s>",
45                     item.get_name(), item.get_type_name(), this.name(),
46                     this.type_name()
47                 )
48             )
49             this.m_item = null;
50             this.m_item.rand_mode( 0 );
51         end
52     endfunction: set_item
53
54 endclass: policy_imp
55
56 `endif

```

E. policy_list.svh

```

1  `ifndef __POLICE_LIST__
2  `define __POLICE_LIST__
3
4  class policy_list implements policy;
5
6      rand POLICY_QUEUE list;
7
8      function new(POLICY_QUEUE policy_queue={});
9          list = policy_queue;
10     endfunction: new
11
12     virtual function string name();
13         int unsigned list_size;
14         int unsigned last_idx;
15         string s;
16
17         list_size = list.size();
18         last_idx = ((list_size) ? (list_size - 1) : 0);

```



```

19         foreach(list[idx]) s = {s, list[idx].name(), ((idx == last_idx) ? "" : ",
20         ")};
21
22         return ({"{", s, "}"});
23     endfunction: name
24
25     virtual function void set_item(uvm_object item);
26         foreach(list[idx]) list[idx].set_item(item);
27     endfunction: set_item
28
29     virtual function void add(policy pcy);
30         list.push_back(pcy);
31     endfunction: add
32
33     virtual function void clear();
34         list.delete();
35     endfunction: clear
36
37     virtual function int unsigned size();
38         return (list.size());
39     endfunction: size
40
41 endclass: policy_list
42
43 `endif

```

APPENDIX B POLICY MACROS

A. *policy_macros.svh*

```

1  `ifndef __POLICY_MACROS__
2  `define __POLICY_MACROS__
3
4  //=====
5  // Embedded POLICIES class macros
6  //=====
7
8  `define start_policies(cls)          \
9  class POLICIES;                      \
10 `LOCAL_POLICY_IMP(cls)
11
12 `define start_extended_policies(cls, parent) \
13 class POLICIES extends parent::POLICIES; \
14 `LOCAL_POLICY_IMP(cls)
15
16 `define end_policies                  \
17 endclass: POLICIES
18
19 `define LOCAL_POLICY_IMP(cls)        \
20     typedef policy_imp#(cls) LOCAL_POLICY_T;
21
22

```

```

23 //=====
24 // Policy template macros
25 //=====
26
27 `include "constant_policy.svh"
28 `include "fixed_policy.svh"
29 `include "ranged_policy.svh"
30 `include "set_policy.svh"
31
32 `endif

```

B. constant_policy.svh

```

1  `ifndef __CONSTANT_POLICY__
2  `define __CONSTANT_POLICY__
3
4  // Full policy definition
5  `define constant_policy(POLICY, TYPE, field, constant) \
6  `CONST_POLICY_CLASS(POLICY, TYPE, field, constant) \
7  `CONST_POLICY_CONSTRUCTOR(POLICY)
8
9  // Policy class definition
10 `define CONST_POLICY_CLASS(POLICY, TYPE, field, constant) \
11     class POLICY`_policy extends LOCAL_POLICY_T \
12         typedef TYPE l_field_t; \
13 \
14         constraint c_const_value { \
15             (item != null) -> (item.field == l_field_t'(constant)); \
16         } \
17 \
18         function new(); \
19         endfunction: new \
20 \
21         virtual function string name(); \
22         return ("POLICY(field==constant)"); \
23         endfunction: name \
24     endclass: POLICY`_policy
25
26 // Policy constructor definition
27 `define CONST_POLICY_CONSTRUCTOR(POLICY) \
28     static function policy POLICY(); \
29     POLICY`_policy new_pcy = new(); \
30     return (new_pcy); \
31     endfunction: POLICY
32
33 `endif

```

C. fixed_policy.svh

```

1  `ifndef __FIXED_POLICY__
2  `define __FIXED_POLICY__
3
4  // Full policy definition

```

```

5  `define fixed_policy(POLICY, TYPE, field, RADIX="%0p") \
6  `FIXED_POLICY_CLASS(POLICY, TYPE, field, RADIX) \
7  `FIXED_POLICY_CONSTRUCTOR(POLICY, TYPE, RADIX)
8
9  // Policy class definition
10 `define FIXED_POLICY_CLASS(POLICY, TYPE, field, RADIX="%0p") \
11     class POLICY`_policy extends LOCAL_POLICY_T \
12         typedef TYPE          l_field_t; \
13 \
14         protected TYPE        m_fixed_value; \
15         protected string      m_radix=RADIX; \
16 \
17         constraint c_fixed_value { \
18             (item != null) -> (item.field == l_field_t'(m_fixed_value)); \
19         } \
20 \
21         function new(TYPE value, string radix=RADIX); \
22             this.set_value(value); \
23             this.set_radix(radix); \
24         endfunction: new \
25 \
26         virtual function string name(); \
27             return ({ \
28                 ~"POLICY(field==", \
29                 $sformatf(m_radix, m_fixed_value), \
30                 ~")~" \
31             }); \
32         endfunction: name \
33 \
34         virtual function void set_value(TYPE value); \
35             this.m_fixed_value = value; \
36         endfunction: set_value \
37 \
38         virtual function TYPE get_value(); \
39             return (this.m_fixed_value); \
40         endfunction: get_value \
41 \
42         virtual function void set_radix(string radix); \
43             this.m_radix = radix; \
44         endfunction: set_radix \
45 \
46         virtual function string get_radix(); \
47             return (this.m_radix); \
48         endfunction: get_radix \
49     endclass: POLICY`_policy
50
51 // Policy constructor definition
52 `define FIXED_POLICY_CONSTRUCTOR(POLICY, TYPE, RADIX="%0p") \
53     static function policy POLICY(TYPE value, string radix=RADIX); \
54         POLICY`_policy new_pcy = new(value, radix); \
55         return (new_pcy); \
56     endfunction: POLICY
57
58 `endif

```

D. ranged_policy.svh

```
1  `ifndef __RANGED_POLICY__
2  `define __RANGED_POLICY__
3
4  // Full policy definition
5  `define ranged_policy(POLICY, TYPE, field, RADIX="%0p") \
6  `RANGED_POLICY_CLASS(POLICY, TYPE, field, RADIX) \
7  `RANGED_POLICY_CONSTRUCTOR(POLICY, TYPE, RADIX)
8
9  // Policy class definition
10 `define RANGED_POLICY_CLASS(POLICY, TYPE, field, RADIX="%0p") \
11     class POLICY`_policy extends LOCAL_POLICY_T \
12         typedef TYPE          l_field_t; \
13 \
14         protected TYPE        m_low; \
15         protected TYPE        m_high; \
16         protected bit         m_inside; \
17         protected string      m_radix=RADIX; \
18 \
19         constraint c_ranged_value { \
20             (item != null) -> \
21                 ((!m_inside) ^ (item >= m_low && item <= m_high)); \
22         } \
23 \
24         function new( \
25             TYPE    low, \
26             TYPE    high=low, \
27             bit     inside=1'b1, \
28             string  radix=RADIX \
29         ); \
30             this.set_range(low, high); \
31             this.set_inside(inside); \
32             this.set_radix(radix); \
33         endfunction: new \
34 \
35         virtual function string name(); \
36             return ({ \
37                 ~"POLICY(field ", \
38                 m_inside ? "inside [" : "outside [", \
39                 $sformatf(m_radix, m_low), \
40                 ", ", \
41                 $sformatf(m_radix, m_high), \
42                 ~"])" \
43             }); \
44         endfunction: name \
45 \
46         virtual function void set_range(TYPE low, TYPE high); \
47             if (low <= high) begin \
48                 this.m_low = low; \
49                 this.m_high = high; \
50             end else begin \
51                 this.m_low = high; \
52                 this.m_high = low; \
```

```

53         end
54     endfunction: set_range
55
56     virtual function TYPE get_low();
57         return (this.m_low);
58     endfunction: get_low
59
60     virtual function TYPE get_high();
61         return (this.m_high);
62     endfunction: get_high
63
64     virtual function void set_inside(bit inside);
65         this.m_inside = inside;
66     endfunction: set_inside
67
68     virtual function bit get_inside();
69         return (this.m_inside);
70     endfunction: get_inside
71
72     virtual function void set_radix(string radix);
73         this.m_radix = radix;
74     endfunction: set_radix
75
76     virtual function string get_radix();
77         return (this.m_radix);
78     endfunction: get_radix
79 endclass: POLICY``_policy
80
81 // Policy constructor definition
82 `define RANGED_POLICY_CONSTRUCTOR(POLICY, TYPE, RADIX="%0p")
83     static function policy POLICY(
84         TYPE    low,
85         TYPE    high=low,
86         bit     inside=1'b1,
87         string  radix=RADIX
88     );
89         POLICY``_policy new_pcy = new(low, high, inside, radix);
90         return (new_pcy);
91     endfunction: POLICY
92
93 `endif

```

E. set_policy.svh

```

1  `ifndef __SET_POLICY__
2  `define __SET_POLICY__
3
4  // Full policy definition
5  `define set_policy(POLICY, TYPE, field, RADIX="%0p")
6  `SET_POLICY_CLASS(POLICY, TYPE, field, RADIX)
7  `SET_POLICY_CONSTRUCTOR(POLICY, TYPE, RADIX)
8
9  // Policy class definition
10 `define SET_POLICY_CLASS(POLICY, TYPE, field, RADIX="%0p")

```

```

11 class POLICY`~_policy extends LOCAL_POLICY_T
12     typedef TYPE         l_field_array_t[];
13
14     protected l_field_array_t    m_values;
15     protected bit                m_inside;
16     protected string             m_radix=RADIX;
17
18     constrant c_set_value {
19         (item != null) ->
20             ((m_inside) ^ (item.field inside {m_values})));
21     }
22
23     function new(
24         l_field_array_t values,
25         bit             inside=1'b1,
26         string          radix=RADIX
27     );
28         this.set_values(values);
29         this.set_inside(inside);
30         this.set_radix(radix);
31     endfunction: new
32
33     virtual function string name();
34         string values_str = "";
35
36         foreach(m_values[i])
37             values_str = {
38                 values_str,
39                 $sformatf(m_radix, m_values[i]),
40                 i == m_values.size()-1 ? "" : ", ";
41
42         return ({
43             ~"POLICY(field ",
44             m_inside ? "inside {" : "outside {" ,
45             values_str,
46             ~"}~"
47         });
48     endfunction: name
49
50     virtual function void set_values(l_field_array_t values);
51         this.m_values = values;
52     endfunction: set_values
53
54     virtual function l_field_array_t get_values();
55         l_field_array_t l_array;
56         foreach (this.m_values[i])
57             l_array[i] = this.m_values[i];
58         return (l_array);
59     endfunction: get_values
60
61     virtual function void set_inside(bit inside);
62         this.m_inside = inside;
63     endfunction: set_inside
64

```

```

65     virtual function bit get_inside();
66         return (this.m_inside);
67     endfunction: get_inside
68
69     virtual function void set_radix(string radix);
70         this.m_radix = radix;
71     endfunction: set_radix
72
73     virtual function string get_radix();
74         return (this.m_radix);
75     endfunction: get_radix
76 endclass: POLICY``_policy
77
78 // Policy constructor definition
79 `define SET_POLICY_CONSTRUCTOR(POLICY, TYPE, RADIX="%0p")
80     typedef TYPE POLICY``_array_t[];
81     static function policy POLICY(
82         POLICY``_array_t values,
83         bit                inside=1'b1,
84         string              radix=RADIX
85     );
86         POLICY``_policy new_pcy = new(values, inside, radix);
87         return (new_pcy);
88     endfunction: POLICY
89
90 `endif

```

REFERENCES

- [1] J. Dickol, "SystemVerilog Constraint Layering via Reusable Randomization Policy Classes," DVCon, 2015.
- [2] J. Dickol, "Complex Constraints: Unleashing the Power of the VCS Constraint Solver," SNUG Austin, September 29, 2016.
- [3] K. Vasconcellos, J. McNeal, "Configuration Conundrum: Managing Test Configuration with a Bite-Sized Solution," DVCon, 2021.
- [4] C. McClish, "Bi-Directional UVM Agents and Complex Stimulus Generation for UDN and UPF Pins," DVCon, 2021.
- [5] J. Vance, J. Montesano, M. Litterick, J. Sprott, "Be a Sequence Pro to Avoid Bad Con Sequences," DVCon, 2019.